

# 中文显示广告牌

设计者：芮文彬

2004 年 6 月 5 日

## 一、功能简介

该作品显示长度为两个半字，能够显示三种颜色（红、绿、橙）；存储 245 个汉字（490 个字符）。显示方式为左移。

## 二、结构

该显示屏由显示模块、控制模块、行驱动模块、列驱动模块、ISP 下载线接口、电源组成。

- **显示模块：**由 10 块 8\*8 的双色点阵组成；

- **控制模块：**用 89S52 作为控制核心（有 8K 的 FLASH ROM）；

- **行驱动电路：**由两片 74LS138 级联构成 4 线-16 线选及三极管驱动放大电路构成（三极管采用两个中功率管及一个大功率管级联构成）；

- **列驱动电路：**由两组(红色一组、绿色一组) 74HC164 级联（5 个）构成，并由中功率三极管放大。

- **ISP 下载线接口：**是用下载程序的，方便显示内容的更改。

- **电源：**是用开关电源来供电的

## 三、汉字的修改

在电脑中输入想要显示的内容，通过字模转换软件提取字模。将字模复制到程序中，更改程序中的字符数，重新编译。然后通过 ISP 下载线直接下载到单片机里面。（更改 200 个字整个操作过程约需 5 分钟）

注：该方法虽落后，但仍能快速更改内容。还是可暂时用住先的。


## 四、性能

该显示屏刚调试成功时试过连续运行 4 天也很稳定。现在经过半年的折磨（只有其中一行的功放管不太行了），目前善能正常使用其性能应该还算可以。

## 五、指标

- 工作电压+5V
- 最大电流 3A
- 最大功耗 15W
- 使用+5V\5A 开关电源供电

## 六、程序流程图



## 七、程序清单

```

*****
,*                               中文显示广告牌程序                               *
,*                               amygse                                           *
,*                               2004.6.5                                         *
,*                               24MHz 晶振                                       *
,*                               *****
;
;21H~26H--显示内容存储单元
;B  ----每次移入内存单元的数据要左移的次数
;27H-----寻址偏址
;28H-----每次移入内存单元的数据要左移的次数临时值
;29H-----扫描还剩几行寄存
;2AH-----左移速度
;2CH-----应该选通那一行寄存
;30H,31H-----延时同步修正寄存器
;SP 堆栈入口 40H
RXD1 EQU P1.4 ;红色串行内容发送
TXD1 EQU P1.5 ;红色串行 CP 端
RXD2 EQU P1.6 ;绿色串行内容发送
TXD2 EQU P1.7 ;绿色串行 CP 端
EN1 EQU P3.7 ;EN1 为高电平，关闭显示端口；EN1 为低电平时，开放显示端口
;*****中断入口程序*****
ORG 0000H
LJMP MIAN
ORG 0003H
RETI
ORG 000BH
RETI
ORG 0013H
RETI
ORG 001BH
RETI
ORG 0023H
RETI
;*****主程序*****
MIAN: MOV 20H,#01H ;设开始为模式 1
STRAT: SETB EN1 ;关显示端口
      MOV SP,#40H ;堆栈首址
      MOV R0,#21H ;初使化寄存显示单元
      MOV R1,#12H ;21H~2DH 单元清零
      MOV A,#00H
ST:    MOV @R0,A
      INC R0
      DJNZ R1,ST
      LCALL STR
      MOV A,20H
      RL A
      MOV 20H,A
      JB ACC.3, MIAN ;模式的选择，有 5 种,红，绿，橙，绿+红，红+绿，现选前 4 种。

```

```

        LJMP  STRAT          ;循环显示内容
;*****左移显示汉字程序*****
STR:     MOV     DPTR,#TAB    ;赋查表初址
        MOV     2DH,#1
HH:      MOV     2EH,#78;共有多少个半角字符，一个汉字等于两个字符，个数值为实际字符个数减 5
HHH:     LCALL  PLAY          ;运行 PLAY 一次后，内容左移显示八次,寻址偏址不变.
        CLR     C
        MOV     A,DPL
        ADD     A,#10H
        MOV     DPL,A
        MOV     A,DPH
        ADDC    A,#00H
        MOV     DPH,A
        DJNZ    2EH,HHH
        DJNZ    2DH,HH
        RET
;*****左移一个字符程序*****
PLAY:    MOV     B,#01H      ;每次移入内存单元的数据要左移次数次值
NLINE:   MOV     2AH,#01h    ;左移速度
GG:      LCALL  LINE16        ;扫描显示 1 帧
        DJNZ    2AH,GG
        INC     B
        MOV     A,#09H
        CJNE    A,B,NLINE    ;如果 B 中的数超过 8 就完成 1 个字符的显示，否则继续循环.
        RET
;*****扫描显示 1 帧行子程序*****
LINE16:  MOV     29H,#10H    ;扫描的行数
FF:      LCALL  LINE1        ;扫描显示 1 行子程序
        DJNZ    29H,FF
        MOV     A,27H
        SUBB    A,#10H
        MOV     27H,A
        RET
;*****扫描显示一行子程序*****
LINE1:   MOV     28H,B       ;显示内容要左移几次数再显示的赋值给 28H 单元
        LCALL  SEND          ;发送串行口子程序
        LCALL  XH            ;选通行子程序
        CLR    EN1           ;开显示端口,显示延时开始
        LCALL  DL800US       ;延时显示 1ms
        LCALL  MSTR          ;将显示内容移入内存单元 21H~26H,跳出后 27H 加 1
EE:      LCALL  ZY            ;左移速度同步修正
        DJNZ    28H,EE
        LCALL  ZYXZ          ;左移速度同步修正
        SETB    EN1          ;关显示端口,显示延时结束
        RET
;*****左移速度同步修正子程序*****
ZYXZ:    MOV     30H,B
        MOV     A,#09H

```

```

        SUBB  A,30H
        MOV   31H,A
AA:      LCALL  DL22US
        DJNZ  31H,AA
        RET

```

;\*\*\*\*\*显示内容移入内存单元子程序\*\*\*\*\*

```

MSTR:    MOV    R0,#21H
          MOV    R4,#06H
          MOV    A, 27H
BB:      MOVC   A, @A+DPTR
          MOV    @R0,A
          MOV    A,27H
          ADD    A,#10H
          MOV    27H,A
          INC    R0
          DJNZ   R4,BB
          MOV    A,27H
          SUBB   A,#5FH
          MOV    27H,A
          RET

```

;\*\*\*\*\*左移一位子程序\*\*\*\*\*

;21H~26H 存入显示内容,

;R0 为显示末址,R2 为移出高位个数

```

ZY:      MOV    R0, #26H    ;显示单元末址
          MOV    R2, #6H    ;移出高位个数
          CLR    C
CC:      MOV    A,  @R0
          RLC    A          ;将后一字节的最高位移到 C 里，再移到前一字节的最低位
          MOV    @R0,A
          DEC    R0
          DJNZ   R2,  CC
          CLR    C
          RET

```

;\*\*\*\*\*发送数据到串口子程序\*\*\*\*\*

```

SEND:    MOV    R0,#25H
          MOV    R5,#5H
LOOP:    MOV    R2,#08H
          MOV    A,@R0
          LCALL  DD
          DEC    R0
          DJNZ   R5,LOOP
          CLR    C
          RET
DD:      JB 00H,DD1
          JB 01H,DD2
          JB 02H,DD3
          JB 03H,DD4

```

```

        JB 04H,DD5
        RET
DD1:    CLR  TXD1
        SETB TXD2
        RRC  A
        MOV  RXD1,C
        SETB TXD1
        SETB TXD2
        DJNZ R2,DD1
        RET
DD2:    CLR  TXD2
        SETB  TXD1
        RRC  A
        MOV  RXD2,C
        SETB TXD2
        SETB TXD1
        DJNZ R2,DD2
        RET
DD3:    CLR  TXD1
        CLR  TXD2
        RRC  A
        MOV  RXD1,C
        MOV  RXD2,C
        SETB TXD1
        SETB TXD2
        DJNZ R2,DD3
        RET
DD4:    CLR  TXD1
        CLR  TXD2
        RRC  A
        MOV  RXD1,C
        CPL  C
        MOV  RXD2,C
        SETB TXD1
        SETB TXD2
        DJNZ R2,DD4
        RET
DD5:    CLR  TXD1
        CLR  TXD2
        RRC  A
        MOV  RXD2,C
        CPL  C
        MOV  RXD1,C
        SETB TXD1
        SETB TXD2
        DJNZ R2,DD5
        RET

```

\*\*\*\*\*选通行子程序\*\*\*\*\*

```

XH:    MOV    2CH,29H    ;29H 存运行剩几行数值
      MOV    A,#0FH
      SUBB   A,2CH        ;16 列减剩余行数得现在应该扫描行
      MOV    P1,A        ;送 P1 口
      RET

```

;\*\*\*\*\*800US 延时子程序\*\*\*\*\*

```

DL800US: MOV    R6,#1dH
DL1:    MOV    R5,#18H
DL2:    DJNZ   R5,DL2
      DJNZ   R6,DL1
      RET

```

;\*\*\*\*\*22US 延时子程序\*\*\*\*\*

```

DL22US: MOV    R6,#19
DL3:    DJNZ   R6,DL3
      NOP
      RET


```

```

TAB:    DB    "字模"        ;该处写入要显示的字的字模
      END

```

## 八、总原理图:



九、作品照片：

